

"This is a book for everyone who uses Vlookup" - Dr. Nitin

the vlookup book

*Learn how to answer any question about
your data using Microsoft Excel lookup functions*



CHANDOO

The vlookup book

By
Chandoo

<http://chandoo.org/>

© 2013 – All rights reserved

The Vlookup Book

© 2013 Chandoo.org & M/s Purnachandra Rao Duggirala

Self-Publishing

(Contact thru <http://chandoo.org/wp/contact/> or <http://chandoo.org/wp/about/>)

ALL RIGHTS RESERVED. This book contains material protected under International and Federal Copyright Laws and Treaties. Any unauthorized reprint or use of this material is prohibited. No part of this book may be reproduced or transmitted in any form or by any means, electronic or mechanical, including photocopying, recording, or by any information storage and retrieval system without express written permission from the author / publisher.

The information in this book is provided on as-is basis. The authors and publisher shall have neither liability nor responsibility to any person or entity with respect to any loss or damages arising from the information contained in this book.

Author: Chandoo (Purnachandra Rao Duggirala)

Cover Design: Chandoo.org

Published by: Self-publishing

First Published in: October 2013

Acknowledgements

Ian Huitson & Robert Cooke - for the technical editing & proof-reading support. I am very thankful to you for all that you do at Chandoo.org

Harish, Duncan, Ardith, M Armstrong, Eliahu A & others – for sending me mistakes, errors and omissions in first edition. Thank you.

Ravindra – for proof reading the book and making sure it is fit to go. Thanks Ravindra for your support.

Contributors on Chandoo.org – I learned some of the ideas & techniques presented in this book thru commenters on my blog. So many thanks to the ones who helped me learn. Special shout out to, Michael, Lukas, Prem, Rohit, John, Godzilla, Chris, Judgepax, Dan L, Vinod, Vipul, Ayush Jain, Spotpuff, Glen Feechan, Dominik Petri and Bryan. These wonderful people have contributed to my VLOOKUP week during 2010 which eventually gave birth to this book. Thank you.

Jo, Nakshatra & Nishanth – for loving me and making my life so much more beautiful. I love you.

Table of contents:

Who is Chandoo & what is Chandoo.org?	6
A note about downloadable workbook	8
Oh! Why should you bother reading this book?!?.....	9
What is VLOOKUP?.....	10
What about LOOKUP, HLOOKUP?.....	15
Using VLOOKUP with Excel Tables	18
Wild card searches with VLOOKUP	22
Let's mix VLOOKUP with data validation for fun & wow factor	24
Look right, look left... oh forget!	28
What is the sales of Joseph in North for product "Fast car" in Jan 2013?.....	32
I want to lookup my cake and eat it too!.....	40
2 way lookups won't give you hiccups.....	44
2 nd time lucky? No... how about third time?	49
Lookup tips & gotchas.....	53
How to learn more about Excel?	55
Appendix 1: INDEX formula explained	57
Appendix 2: Excel 2013 data model.....	58

Who is Chandoo & what is Chandoo.org?

If you have never looked up Chandoo.org in your internet searches, hello to you :)

My name is Chandoo. My actual name is Purnachandra Rao Duggirala, but all my friends call me Chandoo. It is my nick name and I used the same when registering a website back in 2004.

At Chandoo.org our goal is simple. ***We want you to become awesome in Excel.***

How do we do that?

We do it thru 4 ways.

1. Give Awesome Tips, Tutorials, Examples & Downloads

3 or 4 times every week, we write about various creative & productive ways in which you can use Excel to become awesome at what you do.

You can get these articles right in to your inbox by joining our free e-mail newsletter. Or you can subscribe to our RSS feed & read the articles in your favorite news reader.

[When you join our newsletter](#), you also get a free e-book with 95 excel tips.

Visit <http://chandoo.org/wp/subscribe/> to join our free newsletter.

2. Conduct Awesome Training Programs

We conduct various Excel training programs, all aimed to improve your skills & make you a hero in your office. To date, we have trained more than 5,000 professionals from all parts of the world and made them awesome in Excel.

All our programs are completely online & you can enroll at any time. You can access the training videos 24x7 and learn at a pace that works for you.

Visit <http://chandoo.org/wp/training-programs/> to know more about our various training programs.

3. Offer Excel Tools to Make You Awesome

We sell Excel templates for awesome project management. This product is crafted with so much passion. More than 3,000 customers have bought these from us and have enhanced their productivity and became rock-stars in front of their bosses & colleagues.

Visit <http://chandoo.org/pmt/pmt-index-1.html> to know about our PM Templates.

4. Run an Awesome Excel Forum

Almost 4 years ago, we started an Excel forum. It has been growing steadily and now hosts more than 12,000 discussions with 7,000+ active users. Dedicated users like Hui, Luke, Pablo, Narayan, Bob and Faseeh contribute regularly and answer questions with passion & kindness. It has become a hidden treasure of knowledge, new ideas & learning for many. You too can join our forums & share your knowledge (or ask your questions).

[Register on Chandoo.org forums](#)

Thank you

Once again welcome to the world of Chandoo.org and I wish you an awesome journey ahead.

PS: Since our story is quite long and detailed, I am going to point you to [our welcome page for more](#).

A note about downloadable workbook

Trying to understand Excel formulas without using Excel is like shaving a monkey's head while hanging upside down from a tree branch. It can be done, but very few would survive (the monkeys I mean). So the best way to learn is to open Excel and practice as you learn. That is why we have an example workbook for every lesson. To keep things really simple & awesome, I have combined all the workbooks in to one file. Download it from below link.

Download Excel Workbook.

[Alternative download link](#)

Or just visit - [**http://files.chandoo.org/tvb/examples.xlsx**](http://files.chandoo.org/tvb/examples.xlsx)

Oh! Why should you bother reading this book?!?

In other words, Introduction to this book

Well, that seems like an irrelevant question, given that you have already purchased a copy :), but better late than never.

The purpose of this book is to make you *really good* in answering any data lookup related question using Excel.

It captures various techniques, ideas and formula concepts that are necessary to write lookup formula for almost any situation. It is more of a “how-to” book than a theoretical book. Each chapter starts with a goal or a problem and we work thru various solutions to reach our end point. Along way you will be picking up some useful Excel functions, formula writing techniques and some bad humor.

More than VLOOKUP or any other lookup formulas, I hope this book also enlightens you on how to think about your data problems. I believe that Excel is like a Lego kit. Creative, beautiful & inspiring stuff can only be made when you combine different pieces. If you just take a 2x2 block, you get nothing. Combine it with something else, and you just made progress. *So I hope this book gives you some new building blocks that will enhance your Excel knowledge and make you even more awesome.*

Wishing you all the best.

What is VLOOKUP?

I often tell my excel school students that learning VLOOKUP formula will change your basic approach towards data. You will suddenly feel that you have discovered a superman cape in your attic. It is that awesome.

What does VLOOKUP really do?

Imagine you have a list of data like this

Sales Person	No. Customers	Net Sales	Profit / Loss
Joseph	8	1,592	563
John	8	1,088	397
Josh	8	1,680	753
Jamie	9	2,133	923
Jackie	10	1,610	579
Johnson	10	1,540	570
Jonathan	7	1,316	428
Jagjit	7	1,799	709
Jairam	8	1,624	621
Jessy	6	726	236
Javed	9	2,277	966
Jimmy	6	714	221
Juno	9	2,682	1,023

Now, how do you answer the question – “How many sales did Jimmy make?”

Yes, your guess is right. VLOOKUP is one of the formulas you can use to answer questions like this.

VLOOKUP searches a list for a value in the left most column and returns corresponding value from adjacent columns.

So, in our case, we need VLOOKUP to search for Jimmy and return the amount of sales he made from column 3.

VLOOKUP Syntax & Examples:

The syntax of VLOOKUP is simple:

=VLOOKUP(this value, your data table, column number, optional is your table sorted?)

Here are a few examples to get you started:

Let's say you have some sales data in the range B5:E17. Column B contains Sales person name and C, D & E contain No. of customers, Sales & Profits respectively.

Formula: VLOOKUP("John",B\$5:\$E\$17,3,FALSE)

What it does: Finds Sales for John.

Answer: 1088

Notes: Since Sales value is third column, we use 3.

Formula: VLOOKUP("Jamie",B\$5:\$E\$17,2,FALSE)

What it does: Finds number of customers for Jamie.

Answer: 9

Notes: Customers is in second column.

Formula: VLOOKUP("Jessy",B\$5:\$E\$17,4,FALSE)

What it does: Finds Profits for Jessy.

Answer: 236

Notes: Profit is 4th column.

Formula: VLOOKUP("Joshua",B\$5:\$E\$17,3,FALSE)

What it does: Finds Sales for Joshua.

Answer: #N/A

Notes: In this sample data, we do not have Joshua. So VLOOKUP returns #N/A..

What is FALSE doing at the end of my VLOOKUP formulas?

The first 3 parameters of VLOOKUP are rather simple to understand.

1. What is the value you are looking for
2. Address of your table or range
3. Which column you want to extract

The last parameter is somewhat tricky.

You see, for all the smart things computers (or Excel in particular) can do, they are rather dumb. They need a little bit of assistance from you to do their job better.

In case of VLOOKUP, it wants to know whether your table is sorted by first column or not. If it is sorted (in ascending order), then VLOOKUP will jump roughly to the place where your lookup value could be and starts looking there. This will save VLOOKUP a lot of time.

Think of finding the meaning of a word like **humdinger** in a dictionary. You would not start reading each word from A. Instead, you would jump to page starting with H, then Hu and then Hum and then start looking for **humdinger**. Once you find the word, then you will get the meaning. (Btw, Humdinger means a person or thing or action of remarkable excellence. You could say VLOOKUP is a humdinger feature of Excel)

This is exactly how VLOOKUP will behave when your data is sorted.

Except, you need to tell it first.

This is what that the 4th parameter does. It will tell VLOOKUP what kind of data we have.

If 4th parameter is TRUE or omitted, VLOOKUP assumes that you have a sorted list and tries to do a match accordingly.

Examples of VLOOKUP approximate match and when to use it

Very few business situations demand *approximate answer*. You never hear your boss shout, **“I want to know how much we made in July. Give me an approximate answer?”** Well, you may have a boss who asks for approximate numbers. But when it comes reporting or analysis, starting with approximate numbers is no good.

That is why VLOOKUP approximate match might seem like a useless feature. But wait until you see how it can help.

Example: How much commission to pay?

Let's say you are managing the sales department. And you have the below incentive plan.

Units Sold	% commission
0 to 100	10%
101 to 500	15%
501 to 1000	20%
More than 1000	25%

At the end of first month, you received the sales records of all employees and you are looking at,

Sales Person	Units sold
Sam	85
Samantha	89
Sarah	220
Samuel	340
Samir	750
Seth	2100
Selma	1200

Worry not. You do not have to write a messy, long and nested IF formula to calculate commission %s. Instead you can use VLOOKUP approximate match.

First, we need to re-arrange incentive plan table to something like below:

Units Sold	% commission
0	10%
101	15%

501	20%
1001	25%

Once this is in place, you can use VLOOKUP.

Assuming the units sold is in B2 cell and incentive plan is in the range *incentivePlan*, use below VLOOKUP formula.

=VLOOKUP(B2, incentivePlan, 2)

For example, when looking commission percentage for Sam, who sold 85 units, this VLOOKUP formula will return 10% as the approximate match for 85 is 0. *This is because, 0 is the highest value less than 85.*

Likewise, when looking Samir's sales commission, the output would be 20% because 500 is the highest value less than Samir's units sold – 850.

What if the value I am looking up is not there?

Very simple. VLOOKUP gets chocked!

Of course, there is no such thing as chocked VLOOKUP. Instead it just returns **#N/A**. This means the value you are looking for is not in the table.

#N/A ??? Is there way to show some friendly message

Let's agree on something. #N/A is not the best looking due in a room full of spreadsheet cells. Heck, it is not even remotely pleasant to look at. But how to fix it?

Simple, we use **IFERROR()**. This formula transforms ugly errors in to useful messages. For example,

=IFERROR(LOOKUP("Joshua",\$B\$5:\$E\$17,2,FALSE), "Not found")

Would return "Not found" for our sample data.

Caution: IFERROR() function is available only since Excel 2007. This means, if you are running older version of EXCEL, you need to use some other workaround (like ISERROR()).

Caution 2: Since IFERROR() doesn't care about what kind of error it is, you need to be careful when applying this for your formulas. For example, *if there is some other error with your formula, even then IFERROR() would show "Not Found"*.

Tip: If you have Excel 2013 or above, you can use IFNA() to handle only #N/A errors. This will work beautifully for showing "Not found" kind of messages and yet display other errors when they occur.

What if there are 2 items matching..?

The other side of #N/A problem is your list has too many items matching what you are looking for. In this case VLOOKUP simply returns the first matching item in your list.

What should I do if I want the second item?

Oh, very simple. You would arm twist VLOOKUP and threaten it.

Of course I am joking. Neither VLOOKUP has any arms, nor any feelings. But with a little bit of creativity you can get 2nd item, 3rd item or any other matching item from a list. For more on this, see [chapter 13](#).

Homework time! Let's get cracking...

Now that you are familiar with VLOOKUP, take up a few challenges. Solve below problems. Please download example file before proceeding.

1. How many sales for the person in cell G17?
2. Who made more sales - Jamie or Jackie?
3. What is the sale per customer for Jagjit?
4. What is the profit % for person in G17?

What about LOOKUP, HLOOKUP?

VLOOKUP has a few cousins. The most important ones are HLOOKUP and LOOKUP.

First let's talk about HLOOKUP

Imagine your data is sideways. Something like below.

Now, the value you want to lookup (Sales person name) is in a row (instead of a column) and the value you want (for ex. Sales) is in another row.

This is where HLOOKUP comes handy. *HLOOKUP does lookups horizontally.*

Sales Person	Joseph	John	Josh	Jamie	Jackie	Johns
No. Customers	8	8	8	9	10	10
Net Sales	1,592	1,088	1,680	2,133	1,610	1,540
Profit / Loss	563	397	753	923	579	570

So it can find which column a particular sales person and return the corresponding value from other rows.

Syntax of HLOOKUP

HLOOKUP's syntax is identical to VLOOKUP syntax.

=HLOOKUP(This value, Your data table, Row number, Optional is your table sorted?)

Few HLOOKUP Examples

Let's assume our data is in the range C4:O7.

Question: What is the sales of Jamie?

Formula: HLOOKUP("Jamie", C4:O7, 3, false)

What it does? Looks up the first row across columns for "Jamie" and returns the item in 3rd row.

Answer: 2133

Question: What is the profit of Jairam? *

Formula: HLOOKUP("Jairam", C4:O7, 4, false)

What it does? Looks up the first row across columns for "Jairam" and returns the item in 4th row.

Answer: 621

* Jairam is not shown in the image above, but it is part of the data in example workbook.

Question: What is the sales of the person in cell C20?

Formula: HLOOKUP(C20, C4:O7, 3, false)

What it does? Looks up the first row across columns for C20 and returns the item in 3rd row.

Answer: depends on C20 value.

Now let's understand LOOKUP

Unlike VLOOKUP and HLOOKUP, the LOOKUP formula only works when the *lookup range is sorted*. This means, there are very few scenarios in day-to-day business where we can use LOOKUP.

Syntax of LOOKUP

The standard syntax of LOOKUP is,

=LOOKUP(value you are looking up, lookup range, result range)

See below examples to understand this better.

LOOKUP examples

Before looking at examples, please note that these formulas work only on sorted data. For these examples, we take the same sales person data, but sort sales person name column in ascending order first.

Also assume that our data is in the range **B5:E17**.

Data

Sales Person	No. Customers	Net Sales	Profit / Loss
Jackie	10	1,610	579
Jagjit	7	1,799	709
Jairam	8	1,624	621
Jamie	9	2,133	923
Javed	9	2,277	966
Jessy	6	726	236
Jimmy	6	714	221
John	8	1,088	397
Johnson	10	1,540	570
Jonathan	7	1,316	428
Joseph	8	1,592	563
Josh	8	1,680	753
Juno	9	2,682	1,023

This column should be in ascending order

Question: How many sales did John make?

Answer: 1088

Formula: =LOOKUP("John",\$B\$5:\$B\$17,\$D\$5:\$D\$17)

How it works?

This formula looks up John in the range B5:B17 and returns corresponding item in range D5:D17

Question: How many customers did Jamie have?

Answer: 9

Formula: =LOOKUP("Jamie",\$B\$5:\$B\$17,\$C\$5:\$C\$17)

How it works?

This formula looks up Jamie in the range B5:B17 and returns corresponding item in range C5:C17

Question: How many customers did Joshua have?

Answer: 8 ← *incorrect answer*

Formula: =LOOKUP("Joshua",\$B\$5:\$B\$17,\$C\$5:\$C\$17)

How it works?

This formula looks up Joshua in the range B5:B17. Since Joshua is not available, it finds the nearest value before it, ie *Josh* and returns corresponding item in range C5:C17.

A note about HLOOKUP and LOOKUP

Since most business data will be vertical and unsorted, we are going to talk more about VLOOKUP & related techniques in this book. But keep in mind that there will be situations when you have to use either HLOOKUP or LOOKUP.

Using VLOOKUP with Excel Tables

All lookup tables share similar characteristics.

- They have a lookup column, usually the first one.
- They have a bunch of columns which contain data that we want.
- The lookup table keeps growing (example: every month we add new customers, so customer table grows)
- When extracting matching data, we often get more than 1 column. Example: Lookup a customer ID and get their name, address and phone number.

So, when you write a formula like `=VLOOKUP(customer ID, A2:F41, 2, false)` you know that,

- When we add new customers, the formula must be modified.
- We need to make sure that we have not excluded any columns when specifying the table range for VLOOKUP.

There is an excellent solution to address all these common problems. It is called **Excel Tables**.

If you set up your lookup data as a table, then you can use table names & structural references instead of cell references. You do not have to change formulas when you add more data.

Setting up an Excel Table

Assuming you have sales person data as shown in earlier chapters, select any one cell and press CTRL+T. This opens create table dialog. Make sure the range of table is matching with actual data. Also, check if headers are specified. Click OK.

Sales Person	No. Customers	Net Sales	Profit / Loss
Joseph	8	1592	562.772
John	8	1088	396.9024
Josh	8	1680	752.64
Jamie	9	2133	922.7358
Jackie	10	1610	579.117
Johnson	10	1540	569.8
Jonathan	7	1316	427.5684
Jagjit	7	1799	708.806
Jairam	8	1624	621.3424
Jessy	6	726	235.8774
Javed	9	2277	965.6757
Jimmy	6	714	220.983
Juno	9	2682	1023.183



Sales Person	No. Customers	Net Sales	Profit / Loss
Joseph	8	1592	563
John	8	1088	397
Josh	8	1680	753
Jamie	9	2133	923
Jackie	10	1610	579
Johnson	10	1540	570
Jonathan	7	1316	428
Jagjit	7	1799	709
Jairam	8	1624	621
Jessy	6	726	236
Javed	9	2277	966
Jimmy	6	714	221
Juno	9	2682	1023

Tada... your table is ready.

Alternative way to insert a table:

You can also go to Insert ribbon and choose Insert Table to create a table.

What just happened?

When you created a table, suddenly a slew of things happened to your data. First up, the data changed from raw format to a neat structure.

Alternative rows are shaded in different color so that you can easily read the data. Excel also applied filters so that you can easily filter data you want.

But the most important change is, now that your data is a table, you can refer to it in your formulas using *structural references*.


So instead of writing C5:C17 to access No. of customers column, you can write

My_Table_Name[No. Customers]

That is right! Tables let you talk to your data in a meaningful, plain English fashion.

Using Structural References – A Short De-tour

Before using structural references, let us give our table a meaningful name. For this, go to Design ribbon. This ribbon is shown only when you select a cell inside a table (see below).



Give the table a sensible name like **sales** using top-left corner, table name feature.

Now that our table is called **Sales**, we can use this name anywhere.

Few Examples:

To find out the total sales you can use,

=SUM(Sales[Net Sales])

To find out Average Profit per sales person, you can use,

=AVERAGE(Sales[Profit / Loss])

Learning more about Tables & Structural References

Since the focus of our book is VLOOKUP, let us not talk too much about tables. Instead, I am going to leave you with 2 pointers to learn more.

- [Introduction to Excel Tables](#)
- [Introduction to Structural References & Formulas](#)

End of de-tour.

Using tables & structural references in VLOOKUP formulas

The best way to understand how our newly acquired super powers (ie tables & structural references) can help is to take a look at *before and after formulas*.

Before	After
VLOOKUP("John", \$B\$5:\$E\$17,3,FALSE)	VLOOKUP("John", Sales,3,FALSE)
VLOOKUP("Jamie", \$B\$5:\$E\$17,2,FALSE)	VLOOKUP("Jamie", Sales,2,FALSE)
VLOOKUP("Jessy", \$B\$5:\$E\$17,4,FALSE)	VLOOKUP("Jessy", Sales,4,FALSE)
VLOOKUP("Joshua", \$B\$5:\$E\$17,2,FALSE)	VLOOKUP("Joshua", Sales,2,FALSE)

If all I am doing is replacing B5:E17 with Sales, why bother?!?

Now, *that* is a good question. There are 2 good reasons:

1. One of the headaches of using B5:E17 is that, since we may drag down or copy paste the same formula in several places, we must make sure the reference is locked. So we do not write B5:E17, but we write **\$B\$5:\$E\$17**. This will make sure Excel sticks the reference to B5:E17 no matter where you drag / copy / fill this formula.

With a table name like Sales, it is always locked. No need to worry about reference style anymore!

2. The second thing is whenever we add more sales persons to our team, we need to edit the range address and make changes. For example, we add 5 more people, all our VLOOKUP formulas must be edited and they need to point to new address – B5:E22. Likewise, we choose to track No. of products sold as well. We need to edit our formulas to add this extra column as well (ie change to B5:F17).

With tables, we need to make ZERO changes whenever we add rows or columns.

Wow, what else can Tables do?

Tables let you address individual columns of data with a structural reference (as see above). This feature becomes really useful once you learn INDEX & MATCH formulas in later chapters.

Tables also make it easy for 2 way lookup situations ([read chapter 12 for more on this.](#))

Starting Excel 2013, you can even use tables to be part of Excel data model so that you can avoid writing lookup formulas when all you want is a linked report from different tables.

Visit <http://chandoo.org/wp/tag/excel-tables/> for several examples and usages of tables.

Note about Tables in Excel 2003:

Do you know that tables are available in Excel 2003 under a different name?

Yes, they are called **Lists**. Although Lists are similar to Tables, Microsoft decided to revamp the entire feature and call them as Tables starting Excel 2007.

Wild card searches with VLOOKUP

Often we need our lookup formulas to go wild. Not in the sense of go-wild-and-chomp-a-few-kilo-bytes-of-data sense. But wild like wild cards.

Wildcards eh? Explain them to me.

For eg. In the below data, we may not remember the full name of sales person, but we know that her name starts with *jac*. Now how do you get the sales amount for that person?

Sales Person	No. Customers	Net Sales	Profit / Loss
Joseph	8	1,592	563
John	8	1,088	397
Josh	8	1,680	753
Jamie	9	2,133	923
Jackie	10	1,610	579
Johnson	10	1,540	570
Jonathan	7	1,316	428
Jagjit	7	1,799	709
Jairam	8	1,624	621
Jessy	6	726	236
Javed	9	2,277	966
Jimmy	6	714	221
Juno	9	2,682	1,023

Solution

Simple. Use wild cards. Like this: `=VLOOKUP("jac*", $B$5:$E$17, 3, FALSE)` to fetch the value from 3rd column for the person whose name starts with jac.

This will return the answer as 1,610.

Few more examples

1. Sales for the person whose names starts with **Jac**
`=VLOOKUP("jac*", $B$5:$E$17, 3, FALSE)`

2. Sales for the person whose name as 8 characters
`=VLOOKUP("???????", $B$5:$E$17, 3, FALSE)`

3. Sales for the person whose name ends with **son**
`=VLOOKUP("*son", $B$5:$E$17, 3, FALSE)`

4. Customers for person whose name starts with **ja** and ends with **am**
=VLOOKUP("ja*am",\$B\$5:\$E\$17,2,FALSE)

Keep this in mind

Wild card searches are very useful, but they will find only the first match. So if you search for “Sales of a person whose name starts with **joh**” it will return sales of John alone. Even though Johnson also matches this pattern, VLOOKUP stops after finding first match.

Few wild challenges to keep you busy...

If you think you have a good grip on VLOOKUP’s wild side, try to come up with formulas for below situations.

1. How many sales for the person whose name contains the text in cell A1?
2. Who made more sales - person ending with ph or starting with je?
3. What is the netsales for the person who had 6 customers?

Bonus tip:

Not just VLOOKUP, even COUNTIFS, SUMIFS, COUNTIF, SUMIF, AVERAGEIFS, AVERAGEIF formulas also work with wildcards. So you can count how many customers are from “ABC *” company or what is average salary paid to all employees named “Samantha”.

Let's mix VLOOKUP with data validation for fun & wow factor

The true power Excel is not in individual features like VLOOKUP, Conditional Formatting, Tables or Validations. It lies in the fact that you can combine one feature with another to do something awesome.

For example, you can apply conditional formatting on tables so that any new values typed will be automatically highlighted.

Or you can link form controls to charts to make them interactive.

Or in our case, we can link data validation to VLOOKUP to make it even more powerful.

Enter Bob, with a dull job of looking up and reporting numbers – day in and day out!

To make my point, let's create an imaginary character – Bob. Bob works in a large corporation as invoice analyst. His job involves analyzing thousands of invoices every month and answering questions from all sorts of people – department heads, vendors, bankers, top management, his wife.

An annoying part of Bob's job is 'invoice status calls'. Every day, tens of times people call Bob and ask what the status of invoice number ... xxxxxxxx is. And Bob, for the zillionth time, filters his invoice tracker spreadsheet, finds that invoice number and relays the information to calling party.

One day, frustrated with this repetitive and boring work, Bob approached you. Since you are spreadsheet skilled, smart and kind, you agreed to help.

Let's start.

How invoice data is structured

Here is how the invoice data looks.

Invoice #	Due date	Amount	Raised by	Addressed to	Description	Notes	Paid?	Paid on
INV00001	9-Oct-13	\$ 24,600	Gamify dot com	Sid Timberland	This invoice needs to be paid immediately	Some notes	No	
INV00002	9-Sep-13	\$ 12,200	Bang Inc.	Sofia Thick	This invoice needs to be paid in a month	Some notes	No	
INV00003	11-Nov-13	\$ 4,900	Elephant Design	Samuel Thomson	This invoice needs to be paid in a month	Some notes	No	
INV00004	1-Nov-13	\$ 17,700	Fancy Family Biz	Soren Thelma	This invoice needs to be paid immediately	Some notes	No	
INV00005	6-Mar-14	\$ 5,500	Bang Inc.	Simon Thames	This invoice needs to be paid immediately	Some notes	No	
INV00006	10-Feb-14	\$ 22,400	Helbertch Associates	Sina Tong	This invoice needs to be paid immediately	Some notes	No	
INV00007	2-Oct-13	\$ 24,400	Helbertch Associates	Sina Tong	This invoice needs to be paid in a week	Some notes	No	
INV00008	20-Dec-13	\$ 26,400	Divided Analysts	Sina Tong	This invoice needs to be paid in a week	Some notes	No	
INV00009	8-Nov-13	\$ 15,100	Gamify dot com	Siby Thomos	This invoice needs to be paid immediately	Some notes	No	
INV00010	19-Feb-14	\$ 20,000	Abbot Co.	Sony Tobal	This invoice needs to be paid after talking with A	Some notes	No	
INV00011	6-Nov-13	\$ 15,600	Abbot Co.	Sofia Thick	This invoice needs to be paid immediately	Some notes	No	
INV00012	11-Jan-14	\$ 7,400	Fancy Family Biz	Sony Tobal	This invoice needs to be paid immediately	Some notes	No	
INV00013	2-Oct-13	\$ 9,000	Elephant Design	Samir Thakur	This invoice needs to be paid in a month	Some notes	No	
INV00014	7-Jan-14	\$ 26,200	Helbertch Associates	Samuel Thomson	This invoice needs to be paid after talking with H	Some notes	No	

As you can see, invoice number is in the first column followed by 8 columns of details.

All of this data is in an excel table called ***invoices***.

Simplifying Bob's life – the vision

Here is how we can simplify Bob's life.

1. Bob gets a call for 'invoice status'
2. Bob selects the invoice number from a drop down.
3. The said invoice's details show up in a neat table.
4. Bob reads out the details (or can even print them for the caller).
5. Bob is happy!

Implementing this vision

Step 0: Create a blank sheet for showing output

Just create a blank sheet where Bob can select invoice ID and see its details. Something like this:

Invoice Details: INV00010			
Due date	19-Feb-14	Amount	\$ 20,000.00
Raised by	Abbot Co.		
Addressed to	Sony Tobal		
Description	This invoice needs to be paid after talking with Abbot Co.		
Notes	Some notes		
Paid?	No	Paid on	


Step 1: Setting up invoice selection drop down.

This can be done by data validation feature of Excel. To set up a data validation drop down, we first need to have a list of all possible invoice numbers.

Since all the invoices are in **invoices** table, we can just refer to invoices [Invoice #] to access all the invoice numbers.

Unfortunately, a structural reference like this will not work in data validation feature. So we will have to create a name (using define name).

1. Select invoice ID column in invoices table (you can quickly select entire column by placing mouse pointer on column header and clicking when mouse pointer turns to black colored downwards arrow).
2. Go to Formula ribbon and click Define name




3. Give the name as something like *IstInvoices* and click ok.
 4. Now that the list of numbers is ready, select the invoice ID selection cell (let's say D3) in our output sheet.
 5. Go to Data Ribbon > Data Validation
 6. Select List type of validation and type *=IstInvoices* as the list name.
- Tip: You can press F3 to pull up paste names box and select the name instead of typing.



7. Now Bob can select an invoice ID from list of invoices using drop down.

Select an Invoice



Step 2: Writing the VLOOKUP formulas

Please keep in mind that invoice selection is tied to cell D3.

So, once an invoice is selected, we want to know when it is due. Invoice Due date is in 2nd column of our *invoices* table. So we will write below VLOOKUP formula.

```
=VLOOKUP(D3, invoices, 2, false)
```

Likewise, to get Amount, Raised by, Addressed to, Description, Notes, Paid? And Paid on details, we will use column numbers 3,4,5,6,7,8 and 9 respectively.

Step 3: Bob is free

There is no step 3. Bob is finally free. Now he can answer any invoice related questions with a simple click of mouse.


Ideas to keep in mind

We did not invent a character like Bob to just understand this technique alone. I want you to keep these important ideas in your mind and apply them whenever you can.

- Combine features of Excel to do cool, interesting things.
- Use features like data validation to make your & your user's life simple.
- Use features like tables & structural references to write simpler formulas.

Look right, look left... oh forget!

VLOOKUP cannot go left!



Xxx	Yyy	Zzz	Abcd	Efg
Xxx	Xxx	Xxx	Xxx	Xxx
Xxx	Xxx	Xxx	Xxx	Xxx
Xxx	Xxx	Xxx	Xxx	Xxx
Xxx	Xxx	Xxx	Xxx	Xxx
Xxx	Xxx	Xxx	Xxx	Xxx
Xxx	Xxx	Xxx	Xxx	Xxx
Xxx	Xxx	Xxx	Xxx	Xxx
Xxx	Xxx	Xxx	Xxx	Xxx
Xxx	Xxx	Xxx	Xxx	Xxx
Xxx	Xxx	Xxx	Xxx	Xxx
Xxx	Xxx	Xxx	Xxx	Xxx
Xxx	Xxx	Xxx	Xxx	Xxx
Xxx	Xxx	Xxx	Xxx	Xxx
Xxx	Xxx	Xxx	Xxx	Xxx
Xxx	Xxx	Xxx	Xxx	Xxx

VLOOKUP can lookup the green column and get value from red one, but not the other way around. For that we can use INDEX + MATCH combination.

There is no argument that VLOOKUP is a beautiful & useful formula. But it suffers from one nagging limitation. ***It cannot go left****.

* Of course, with a bit of clever formula magic anything is possible, including making VLOOKUP go left. [See this excellent article](#) to know how to do this.

Back to the cannot go left problem

Let me explain, Imagine you have data like below. Now, if you want to find-out who made \$2,133 in sales, there is no way VLOOKUP can come to rescue. This is because, once you search a list using VLOOKUP, you can only return corresponding items from the column at right, not at left.

Sales Person	No. Customers	Net Sales	Profit / Loss
Joseph	8	1,592	563
John	8	1,088	397
Josh	8	1,680	753
Jamie	9	2,133	923
Jackie	10	1,610	579
Johnson	10	1,540	570
Jonathan	7	1,316	428
Jagjit	7	1,799	709
Jairam	8	1,624	621
Jessy	6	726	236
Javed	9	2,277	966
Jimmy	6	714	221
Juno	9	2,682	1,023

One easy fix would be move the sales data to the left of person name. But this is an annoying fix, because, god knows you may want to lookup based on profit values or something else in future. A better alternative is...,

Using INDEX & MATCH formulas

.., to use a formula combination called INDEX + MATCH (or OFFSET + MATCH would work too).

The basic syntax of this combination is like this:

*=INDEX(column with data you want,
MATCH(value you are looking for, column which contains this data,0))*

So, for eg: =INDEX(\$B\$5:\$B\$17,MATCH(1088,\$D\$5:\$D\$17,0)) would find the position of 1088 in list D5:D17 and return corresponding element from B5:B17 (ie the value from left). See more examples below.

Few more INDEX + MATCH examples

Situation: Which person made sales = 1088

Answer: John

Formula: =INDEX(\$B\$5:\$B\$17,MATCH(1088,\$D\$5:\$D\$17,0))

What it does?

The MATCH(1088,\$D\$5:\$D\$17,0) finds at which position 1088 occurs in column D (Sales). Once the position is found, INDEX((\$B\$5:\$B\$17, ...) will return the corresponding sales person name from column B.

Situation: Who made maximum sales?

Answer: Juno

Formula: =INDEX(\$B\$5:\$B\$17,MATCH(MAX(\$D\$5:\$D\$17), \$D\$5:\$D\$17,0))

What it does?

The MAX(\$D\$5:\$D\$17) finds maximum sales value in column D. Then MATCH will find the position of this sales value. Finally INDEX will tell us who is the person generating this sales (from column B).

Situation: Who sold to minimum number of customers?

Answer: Jessy

Formula: =INDEX(\$B\$5:\$B\$17,MATCH(MIN(\$C\$5:\$C\$17),\$C\$5:\$C\$17,0))

What it does?

The MIN(\$F\$5:\$F\$17) finds minimum number of customers in column F. Then MATCH will find the position of this value. Finally INDEX will tell us who is the person generating this sales (from column B).

Situation: What is sale per customer for the person who has lowest profit ratio?

Answer: 119

Formula: =INDEX(\$D\$5:\$D\$17,MATCH(MIN(\$F\$5:\$F\$17),\$F\$5:\$F\$17,0))
/INDEX(\$C\$5:\$C\$17,MATCH(MIN(\$F\$5:\$F\$17),\$F\$5:\$F\$17,0))

What it does?

This is for you to figure out. Take a sip of that coffee and get going.

What is 0 doing at the end of MATCH?

Just like we are using FALSE in VLOOKUP to do *exact lookups*, In MATCH too, we need to tell Excel what kind of a list we have. MATCH is slightly smarter than VLOOKUP. It can search approximate matches when your list is sorted in ascending or descending order.

So instead of true or false, MATCH takes -1, 0 and 1.

Match criteria	What it means?
1	Find approximate match that is less than the lookup value. Your list should be in ascending order.
0	Finds exact match. List can be in any order.
-1	Find approximate match that is more than the lookup value. Your list should be in descending order.

Time for some challenges...

Using the example workbook for this chapter, answer below questions.

1. Who sold second highest?
2. What is the Profit Ratio rank of person who sold second highest?
3. What is difference in sale per customer between the highest selling & lowest selling sales persons?

What is the sales of Joseph in North for product “Fast car” in Jan 2013?

In other words, how to lookup with multiple conditions?

Lookup based on 2 or more conditions



Xxx	Yyy	Zzz	Abcd	Efg
Xxx	Yyy	Zzz	Abcd	Efg
Xxx	Yyy	Zzz	Abcd	Efg
Xxx	Yyy	Zzz	Abcd	Efg
Xxx	Yyy	Zzz	Abcd	Efg
Xxx	Yyy	Zzz	Abcd	Efg
Xxx	Yyy	Zzz	Abcd	Efg
Xxx	Yyy	Zzz	Abcd	Efg
Xxx	Yyy	Zzz	Abcd	Efg
Xxx	Yyy	Zzz	Abcd	Efg
Xxx	Yyy	Zzz	Abcd	Efg
Xxx	Yyy	Zzz	Abcd	Efg
Xxx	Yyy	Zzz	Abcd	Efg
Xxx	Yyy	Zzz	Abcd	Efg
Xxx	Yyy	Zzz	Abcd	Efg

Situation

Not always we want to lookup values based on one search parameter. For eg. Imagine you have data like below and you want to find how much sales Joseph made in North region for product “Fast car” in January 2013?

Data						
Month	Sales Person	Region	Product	No. Customers	Net Sales	Profit / Loss
Jan-13	Joseph	North	FastCar	8	\$ 1,592	\$ 563
Jan-13	Joseph	North	RapidZoo	8	\$ 1,088	\$ 397
Jan-13	Joseph	West	SuperGlue	8	\$ 1,680	\$ 753
Jan-13	Joseph	West	FastCar	9	\$ 2,133	\$ 923
Jan-13	Joseph	West	RapidZoo	10	\$ 1,610	\$ 579
Jan-13	Joseph	Middle	SuperGlue	10	\$ 1,540	\$ 570
Jan-13	Joseph	Middle	FastCar	7	\$ 1,316	\$ 428
Jan-13	Joseph	Middle	RapidZoo	7	\$ 1,799	\$ 709
Jan-13	Lawrence	North	SuperGlue	8	\$ 1,624	\$ 621
Jan-13	Lawrence	North	FastCar	6	\$ 726	\$ 236
Jan-13	Lawrence	North	RapidZoo	9	\$ 2,277	\$ 966
Jan-13	Lawrence	West	SuperGlue	6	\$ 714	\$ 221
Jan-13	Lawrence	West	FastCar	9	\$ 2,682	\$ 1,023
Jan-13	Lawrence	West	RapidZoo	6	\$ 1,500	\$ 634
Jan-13	Lawrence	Middle	SuperGlue	7	\$ 917	\$ 403
Jan-13	Lawrence	Middle	FastCar	7	\$ 1,939	\$ 760
Jan-13	Lawrence	Middle	RapidZoo	6	\$ 1,500	\$ 634

In such cases, what do we do?

Simple, you use your index finger to scan the list for match and get the answer.

Of course, using index finger is not going to work. It is not scalable. Plus you may want to use your index finger for better things, like playing Angry Birds for example.

Time for some new formulas

While VLOOKUP will work in this case, it's time we introduced few more formulas & techniques. Say hello to,

SUMIFS

SUMPRODUCT

SUM

Array use of INDEX + MATCH

Let's examine the approach with each of these formulas. Along way, you will also understand when to use each of these approaches.

Using SUMIFS to do multiple condition lookups

By far, the simplest way to do multiple condition lookups is to use a summing formula, not a lookup formula! Sounds confusing, read on.

You see, SUMIFS formula can sum up values that meet multiple conditions. So you can ask a question like what is the total sales in the **west region** for the product **fastcar** in the month **February 2013**? And SUMIFS can give you the answer.

Take a pause and examine the behavior of SUMIFS.

Does it look similar to our multiple condition lookup problem?

So, if we can ask SUMIFS to sum up sales of **Joseph** in **North region** for the product **fastcar** in the month **January 2013**, then our look up is done.

Before we write the formula, let me say a few assumptions.

The data is in a table – Sales

With below columns – Month, Sales Person, Region, Product, No. of Customers, Net Sales, Profit / Loss.

(Note: In the example workbook, the table is named as Sales2. But we are going to refer to it as Sales to keep it simple)

```
=SUMIFS(Sales[Net Sales],  
Sales[Sales Person], "Joseph",  
Sales[Month], "Jan-2013",  
Sales[Region], "North",  
Sales[Product], "Fastcar")
```

Why are we summing when we want just one value?

Simple, since there will be only one value matching above conditions, SUMIFS will return that value alone. In other words it sums up just one value.

That sounds cool, are there any limitations?

While SUMIFS approach is incredibly cool, it has a few limitations.

1. It works only when you want to fetch a number column (like sales, no. of customers, profits etc.). If you want to fetch text values like 'What is the description of invoice sent by Acme Inc on 23rd of August 2013 to Purchasing Department?' In such cases SUMIFS will not work because it can't add up text values.
2. It works only when the criteria matches one value. Since SUMIFS sums up all values that match given conditions, if your lookup results in multiple matches, then SUMIFS answer will be wrong. When VLOOKUP sees multiple matches, it returns the first one. Whereas SUMIFS just adds all the matching values and gives you the total.
3. SUMIFS formula does not work in Excel 2003 or below versions. I know Excel 2003 is a decade old, but there are many people still using it. So keep in mind that your SUMIFS formula may not work when your client opens the workbook in Excel 2003.

Bonus: [How to use SUMIFS formula?](#)

Using SUMPRODUCT & SUM formulas

Caution:

The discussion on SUMPRODUCT & SUM may be way too advanced for some of you. If you feel lost, skip to next section.

Just in case you cannot use SUMIFS (because you are running older version of Excel or you hate SUMIFS or you are not in the mood), you can use either SUMPRODUCT or array version of SUM formulas to do what SUMIFS is doing.

SUMPRODUCT Version

First take a look at the formula.

```
=SUMPRODUCT(Sales[Net Sales],  
(Sales[Sales Person]="Joseph")*  
(Sales[Month]="Jan-2013")*  
(Sales[Region]="North")*  
(Sales[Product]="Fastcar"))
```

If observe closely, this formula is similar to SUMIFS version. Instead of using list, criteria here we are using list=criteria.

Since explanation of how SUMPRODUCT works and why we have * between (Sales[Sales Person]="Joseph") and (Sales[Month]="Jan-2013") is beyond the scope of this book, I am going to leave you with 2 very useful links on SUMPRODUCT formula. Read them to understand better.

[Introduction to SUMPRODUCT formula](#)

[Advanced examples of SUMPRODUCT formula](#)

Caution:

While SUMPRODUCT is not inflammable, it is a very powerful formula. It has lots of uses and can take several weeks of dedicated practice before you get a hang of it.

My advice is simple. Use SUMPRODUCT only if you have no other option. For multiple condition lookups, your best choice is to use either SUMIFS or INDEX+MATCH or VLOOKUP with helper columns.

SUM Version

Again, take a look at the formula.

```
=SUM(Sales[Net Sales]*  
(Sales[Sales Person]="Joseph")*  
(Sales[Month]="Jan-2013")*  
(Sales[Region]="North")*  
(Sales[Product]="Fastcar"))
```

This formula is almost like SUMPRODUCT formula, except it says SUM instead of SUMPRODUCT.

This is because, for situations like above, both SUM & SUMPRODUCT work exactly same, with one minor difference. For the SUM formula, you need to press CTRL+Shift+Enter after typing it to get the result.

These formulas are called as array formulas. Again, the discussion of array formulas is out of scope for this book. But if you are curious, check out [array formulas page](http://chandoo.org/array-formulas-page) on chandoo.org for some fantastic examples.

Using INDEX + MATCH formulas

While the SUMIFS approach works beautifully when fetching a number column value based on multiple conditions, it fails when trying to get text values. But there is an approach that works well whether getting numbers or text by looking up based on multiple conditions.

It involves using INDEX + MATCH formulas in a creative way.

First the formula...

```
=INDEX(Sales[Net Sales],  
MATCH(valSalesman&valMonth&valRegion&valProduct,  
Sales[Sales Person]&Sales[Month]&Sales[Region]&Sales[Product],0))
```

For the sake of this formula, assume the names valSalesman, valMonth, valRegion & valProduct contain the combination you are trying to look up (viz. Joseph, Jan-2013, North and Fastcar).

Understanding how this formula works:

The core of this formula is MATCH(). See how we are using valSalesman&valMonth&valRegion&valProduct as the value to lookup. This is nothing but a concatenation (& is used to combine text values in to one big text) of all 4 criteria values. It results in a big text like "Joseph41275NorthFastcar".

41275...? Let's detour

You might be wondering what **41275** is doing. Rest all seems to makes sense. You see, Dates are actually numbers in Excel.

Number 1 corresponds to the first date of Excel world. It is 1-Jan-1900.

Number 2 corresponds to the second date – ie – 2nd of January, year 1900.

Likewise 41275 means – 1st of January, 2013. It simply means 1-Jan-2013 is the 41275th day since 1st of Jan 1900.

Dates are numbers eh... that means

Once you realize dates are numbers, your mind untangles a few more mysteries. Key things to realize are,

- Since dates are numbers, to find out what is the date 30 days hence, you just add 30 to the input date.
- Likewise, to know what is the date 3 weeks prior, just subtract 21.
- Since each day is one number, each hour is 1/24.
- Each minute is 1/24/60
- And each second is 1/24/60/59. Only kidding, of course it is 1/24/60/60
- So to say 9:00 AM on 1-Jan-2013, you can also say 41275 + 9/24 (or 41275.375)
- For more on [using dates & times in Excel see this](#).

<End of detour>

Now that you have understood the mystery behind *Joseph41275NorthFastcar* lets focus on the next part of MATCH formula.

Sales[Sales Person]&Sales[Month]&Sales[Region]&Sales[Product]

This part concatenates the 4 columns of sales table in to one big text column.

Now the match formula looks up *Joseph41275NorthFastcar* in this column and tells us where such a match is. In other words, it tells us where sales person is **Joseph**, month is **41275**, region is **North** and product is **Fastcar**. If there are multiple matches like this, MATCH will tell us the location of first match.

Once we know the location, INDEX will return the corresponding netsales amount from Sales[Net Sales] column.

But wait...

Since this formula (MATCH portion) uses concatenated columns, we have to press CTRL+Shift+Enter after typing this formula to get the result.

But can't we use VLOOKUP alone?

While the approaches involving SUMIFS, SUMPRODUCT, SUM, INDEX + MATCH are interesting & useful, there is one more that is very handy and simple to use.

We can use simple VLOOKUP formula with some extra help to get data meeting multiple conditions.

Introducing Helper columns

Helper columns are new columns of data made from existing data to help us meet an objective. *In this case, the objective is to get data that meets multiple conditions.*

The first step is to insert a blank column to the left of our sales table.

In this column, write a formula to concatenate month, sales person, region and product values.
Something like =C5&D5&E5&F5

At this stage your table should look like this:

Data				
helper column	Month	Sales Person	Region	Product
41275JosephNorthFastCar	Jan-13	Joseph	North	FastCar
41275JosephNorthRapidZoo	Jan-13	Joseph	North	RapidZoo
41275JosephWestSuperGlue	Jan-13	Joseph	West	SuperGlue
41275JosephWestFastCar	Jan-13	Joseph	West	FastCar
41275JosephWestRapidZoo	Jan-13	Joseph	West	RapidZoo
41275JosephMiddleSuperGlue	Jan-13	Joseph	Middle	SuperGlue
41275JosephMiddleFastCar	Jan-13	Joseph	Middle	FastCar
41275JosephMiddleRapidZoo	Jan-13	Joseph	Middle	RapidZoo
41275LawrenceNorthSuperGlue	Jan-13	Lawrence	North	SuperGlue
41275LawrenceNorthFastCar	Jan-13	Lawrence	North	FastCar
41275LawrenceNorthRapidZoo	Jan-13	Lawrence	North	RapidZoo

Now, select entire data including this helper column and give it a name like **Sales2**. Although this step is not necessary, it will make the job of writing lookup formulas simpler.

Writing the VLOOKUP formula

Now that we have the **Sales2** ready, simply write below formula.

```
=VLOOKUP(valMonth&valSalesman&valRegion&valProduct, Sales2, 7, FALSE)
```

7 because netsales data is the 7th column in Sales2 range.

When to use which formula?

Now that you have seen a variety of formulas to get what you want, it's time we understood when to use each method.

- Use SUMIFS if you need to extract number data that meets multiple criteria.
- Use INDEX+MATCH approach for text values or mixed values.
- Use VLOOKUP + Helper column if can add helper columns to your data.
- Use SUMPRODUCT or Array SUM versions only as a last resort.

Also, another thing I want to plant in your mind is,

There are many ways to solve your problems using Excel. Often there is no right or wrong way. Just go with whatever works and easy to implement.

I see many analysts wasting hours of time trying to fine tune their formulas. The reality is your bosses or clients never ask for *fastest spreadsheets*. They ask for most usable, accurate and simple ones. So instead of focusing your energies to reduce the length of your formulas or trying to use complex ones, you should spend time impressing your bosses & clients.

Excel is a vehicle, not a destination.

I want to lookup my cake and eat it too!

In other words, How to extract not one but several columns using LOOKUP formulas?

Get Multiple Outputs - VLOOKUP

Xxx	Yyy	Zzz	Abcd	Efgh
Xxxx	Xxxx	Xxxx	Xxxx	Xxxx
Xxxx	Xxxx	Xxxx	Xxxx	Xxxx
Xxxx	Xxxx	Xxxx	Xxxx	Xxxx
Xxxx	Xxxx	Xxxx	Xxxx	Xxxx
Xxxx	Xxxx	Xxxx	Xxxx	Xxxx
Xxxx	Xxxx	Xxxx	Xxxx	Xxxx
Xxxx	Xxxx	Xxxx	Xxxx	Xxxx
Xxxx	Xxxx	Xxxx	Xxxx	Xxxx

Find a match in this column
& return data in these columns

The previous chapter was heavy. You learned how to use various techniques to get data that meets multiple criteria. Now, let me show you something that is cool, simple and very useful.

VLOOKUP is great for extracting information from a huge data table based on what you are looking for. But what if you need to extract more than one column of information? For e.g. let's say you have salesperson's name in left most column, and monthly sales figures in next columns, one for each month. Now, you want to find the total sales made by a given sales person. How do you go about it?

Data												
Sales Person	Jan	Feb	Mar	Apr	May	Jun	Jul	Aug	Sep	Oct	Nov	Dec
Joseph	\$ 22	22	24	25	26	28	30	32	33	33	36	36
John	10	10	11	11	12	12	13	13	14	15	15	15
Josh	26	26	27	27	27	27	29	30	30	32	32	34
Jamie	18	19	20	20	20	21	22	23	23	24	25	25
Jackie	19	20	22	24	25	25	25	27	29	30	30	31
Johnson	16	17	18	18	19	20	22	24	25	26	27	28
Jonathan	15	16	17	17	18	18	18	19	19	20	20	22
Jagjit	10	11	11	11	12	13	13	14	14	14	15	15
Jairam	21	21	21	23	24	24	25	26	27	29	30	32
Je	25	26	27	28	29	30	31	32	33	34	35	36

Using arrays in VLOOKUP to get multiple columns

Simple. You can use an array in the column number field of VLOOKUP. So for eg. =SUM(VLOOKUP(value, list, {2,3,4},false)) will give you sum of values in columns 2,3 and 4. Of course you must press CTRL+SHIFT+Enter after typing the formula.

Few more examples

Assume that all our data is in B5:N17 range (13 columns – 1st column for sales person name and 12 columns for one month each).

Maximum sales per month of Josh

Answer: 34

Formula: =MAX(VLOOKUP("Josh",\$B\$5:\$N\$17,{2,3,4,5,6,7,8,9,10,11,12,13},FALSE))

Explanation:

This formula takes all months data for Josh (ie columns 2 thru 13) and passes them to MAX to find the maximum sales for any month.

Average monthly sale for Jonathan

Answer: 18.25

Formula: =AVERAGE(VLOOKUP("jonathan",\$B\$5:\$N\$17,{2,3,4,5,6,7,8,9,10,11,12,13},FALSE))

Explanation:


This formula takes all the monthly values for Jonathan and passes them to Average.

What is the prior 3 month running total for **Johnson** from the maximum sales month?

Answer: 81

Explanation of the answer:

See this image to understand how we arrived at the total.



Writing the formula

Formula for this is elaborate. So let's break it in to smaller chunks.

Formula 1 – Finding the maximum sales month for Johnson

The first step is to find in what month Johnson reported maximum sales.

For the sake of making our formula generic, let's assume that the name of the person (Johnson) is typed in to a cell – Q25.

So the maximum sales for the person in Q25 becomes,

`MAX(VLOOKUP( Q25 , $B$5:$N$17, {2,3,4,5,6,7,8,9,10,11,12,13}, FALSE))`

Let's keep this maximum value in the cell **R25**.

Now that we know what the maximum value is, we need to find which month (ie which column) this value belongs.

This can be done by our trusty friend MATCH. Remember, MATCH() tells us the position of a value we are looking for in a list of values.

So if we ask `=MATCH(R25, list of all monthly values for the person in Q25, 0)`, we will get the corresponding column number.

The actual formula is,

`=MATCH($R$25, VLOOKUP($Q$25, $B$5:$N$17, {2,3,4,5,6,7,8,9,10,11,12,13}, FALSE), 0) + 1`

I am sure the formula is tricky to understand. So I will give you a few minutes alone time with it. Go ahead and get up close with it.

Hey, wait a sec... what is +1 doing at the end?!?

Remember, match finds the position of the value we are looking for. Since Johnson reported maximum sales in December, we will get answer as 12. But the real column as per our table is 13 (because first column is taken up by sales person name).

So we need to add 1 to this result.

Caution: Both `MAX(VLOOKUP(...))` and `MATCH(R25, VLOOKUP(...), 0) + 1` formulas are array formulas. This means you need to press CTRL+Shift+Enter after typing them to get result in the cell. Keep this maximum month in the cell **Q26**.

Formula 2 – Finding previous 2 months

After a heavy formula, we now have some easy ones. Since we know which month is the maximum sales month, we need to also figure out what the prior 2 months are.

This is easily done by subtracting 1 & 2 from the value in Q26.

So we end up with 3 cells (Q26:S26) like this:

Name	Johnson	28	max sales value
Max sales month	13	12	11
Sum of sales for 3 prior months			

person we are interested in

max sales month and prior 2 month columns numbers

Formula 3 – Find the running total

Now that we have laid foundation, let's write our final formula.

=SUM(VLOOKUP(\$Q\$25, \$B\$5:\$N\$17, \$Q\$26:\$S\$26, FALSE))

This time, instead of saying {2,3,4} or {13,12,11}, we just referred to the range Q26:S26 that contains these numbers. And lo, we have the sum of 3 month sales that we want.

A final look at our spreadsheet.

This is how the whole thing looks:

Question 3: What is prior 3 month running total for Johnson from the maximum sales month?			
Name	Johnson	28	
Max sales month	13	12	11
Sum of sales for 3 prior months	81		

Important lesson for you

The 3rd example (adding up 3 months sales from maximum sales month) is a rather complex one. But I have included in the book for one reason. I want to show you the process behind answering any question your boss asks. Although it seems like there is no one formula that gives the answer, once you start combining basic formulas like VLOOKUP, MATCH, MAX and SUM, you get the answer.

Also, keep in mind that you should break your problem in to smaller chunks and solve one at a time. If you attempt to write one big nested formula to answer this question, you might arrive at a solution very late. Remember, your boss (or client) wanted the result. They don't care how many extra cells you use in this process.

Some homework?

In the downloadable workbook, I have some homework problems. Solve them to sharpen your skills.

2 way lookups won't give you hiccups

Lookup is when you find a value in one column and get the corresponding element from other columns. **2 Way Lookup is when you look up value at the intersection corresponding to a given row & column values.**

In other words, it is 2 lookups mixed in to one. See this example to understand better.

Let's say you have sales data where person's name is in first column and sales for each month is listed in one column, *ie* you have sales for January in 2nd column, February in 3rd column... December in 13th column.

Now you want to know sales of Joseph in March.

To find this value, you need a 2way lookup. First you lookup Joseph in salesperson names column, then you lookup march in months row and then get the value from intersection.

2 Way Lookup Formulas in Excel

Xxxx	Yyy	Zzz	Abcd	EfgH
Xxxx	Xxxx	Xxxx	Xxxx	Xxxx
Xxxx	Xxxx	Xxxx	Xxxx	Xxxx
Xxxx	Xxxx	Xxxx	Xxxx	Xxxx
Xxxx	Xxxx	Xxxx	Xxxx	Xxxx
Xxxx	Xxxx	Xxxx	Xxxx	Xxxx
Xxxx	Xxxx	Xxxx	Xxxx	Xxxx
Xxxx	Xxxx	Xxxx	Xxxx	Xxxx
Xxxx	Xxxx	Xxxx	Xxxx	Xxxx
Xxxx	Xxxx	Xxxx	Xxxx	Xxxx

Lookup the value at intersection by matching both row & column

(c) chandoo.org

So how to write such a formula in Excel?

Writing 2way lookup formulas

While the problem may seem complicated, the solutions to two way lookups are surprisingly simple. Below, we will learn 4 different ways to write 2 way lookup formulas.

Keep this in mind:

I use various named ranges in the below examples.

valSalesPerson and *valMonth* refer to the name of sales person & month we are looking for.
lstSalesPerson and *lstMonth* refer to the list of sales persons (first column) and list of months (first row).
tblData has the sales figures for everyone.

Sales Person	Jan	Feb	Mar	Apr	May	Jun	Jul	Aug	Sep	Oct	Nov	Dec
Joseph	\$ 22	22	24	25	26	28	30	32	33	33	36	36
John	10	10	11	11	12	12	13	13	14	15	15	15
Josh	26	26	27	27	27	27	29	30	30	32	32	34
Jamie	18	19	20	20	20	21	22	23	23	24	25	25
Jackie	19	20	22	24	25	25	25	27	29	30	30	31
Johnson	16	17	18	18	19	20	22	24	25	26	27	28
Jonathan	15	16	17	17	18	18	18	19	19	20	20	22
Jagjit	10	11	11	11	12	13	13	14	14	14	15	15
Jairam	21	21	21	23	24	24	25	26	27	29	30	32
Jessy	25	26	27	28	29	29	29	30	31	34	37	40
Javed	22	24	25	26	28	28	29	31	34	35	35	35
Jimmy	23	24	24	24	25	26	27	28	30	33	35	37
Juno	19	20	20	21	22	24	24	26	27	27	28	30

Technique 1 – Using INDEX & MATCH Formulas

If you know the row number and column number in a given table, you can use INDEX formula to get the element at the intersection. And we can use MATCH formula to find the position of a value in a list.

Combining both,

`=INDEX(tblData,MATCH(valSalesPerson,lstSalesPerson,0),MATCH(valMonth,lstMonth,0))`

is the formula we use to get the sales amount of *valSalesPerson* for *valMonth*.

How this works?

We will understand this formula from inside out.

MATCH(valSalesPerson,lstSalesPerson,0) will find what is the position of *valSalesPerson* in the *lstSalesPerson* list. So when you look up **Joseph**, this is going to tell you the position (ie row number) of *Joseph* in the list of sales persons.

MATCH(valMonth,lstMonth,0) will find the position of *valMonth* in the *lstMonth*. This result is just like above. So it is going to tell us the column number of **March**.

INDEX(tblData, MATCH(...), MATCH(...)) will then give us the value at the intersection of row & column numbers found by first & second MATCH formulas.

Technique 2 – Using Named Ranges & Intersection (SPACE) Operator

Do you know that you can write =range1 range2 to get the value(s) at the intersection of range1 & range2? That is right, excel has an intersection operator. Since this technique is somewhat tricky to understand with written text, I made a short video. Watch it on YouTube.

<http://www.youtube.com/watch?v=kF73uvOhA4>

However, you need to create named ranges for your data all the time. A simpler alternative is to use Excel Tables feature so the names are created for you automatically. This feature is available since Excel 2007.

Technique 3 – Using SUMPRODUCT Formula

This is an absolute beauty.

You can use SUMPRODUCT to get the value at intersection like this:

=SUMPRODUCT((IstSalesPerson=valSalesPerson)*(IstMonth=valMonth),tblData)

How does it work? Simple, When you write (IstSalesPerson=valSalesPerson)*(IstMonth=valMonth), SUMPRODUCT generates a lot of zeros and a one at the intersection. When you use tblData as second argument, the result is value at intersection.

See this illustration to understand the SUMPRODUCT magic.

Step 1: Multiplying a bunch of row values with column values

	FALSE	FALSE	TRUE	FALSE	FALSE	FALSE	FALSE	FALSE	FALSE	FALSE	FALSE	FALSE
FALSE												
FALSE												
FALSE												
FALSE												
TRUE												
FALSE												
FALSE												
FALSE												
FALSE												
FALSE												
FALSE												
FALSE												
FALSE												
FALSE												



Step 2: Result of this multiplication

	FALSE	FALSE	TRUE	FALSE	FALSE	FALSE	FALSE	FALSE	FALSE	FALSE	FALSE	FALSE
FALSE	0	0	0	0	0	0	0	0	0	0	0	0
FALSE	0	0	0	0	0	0	0	0	0	0	0	0
FALSE	0	0	0	0	0	0	0	0	0	0	0	0
FALSE	0	0	0	0	0	0	0	0	0	0	0	0
TRUE	0	0	1	0	0	0	0	0	0	0	0	0
FALSE	0	0	0	0	0	0	0	0	0	0	0	0
FALSE	0	0	0	0	0	0	0	0	0	0	0	0
FALSE	0	0	0	0	0	0	0	0	0	0	0	0
FALSE	0	0	0	0	0	0	0	0	0	0	0	0
FALSE	0	0	0	0	0	0	0	0	0	0	0	0
FALSE	0	0	0	0	0	0	0	0	0	0	0	0
FALSE	0	0	0	0	0	0	0	0	0	0	0	0
FALSE	0	0	0	0	0	0	0	0	0	0	0	0
FALSE	0	0	0	0	0	0	0	0	0	0	0	0

Result

Since this result array has lots of ZEROs and a single ONE, it will give you the value at intersection when multiplied (thru SUMPRODUCT of course).

[Learn more about SUMPRODUCT:](#)

SUMPRODUCT is an advanced Excel formula with lots of uses. It is somewhat difficult to understand too. I suggest you to go thru below links to learn more.

[Introduction to SUMPRODUCT formula](#)
[Advanced examples of SUMPRODUCT formula](#)

Technique 4 – Using VLOOKUP with MATCH Formula

One more creative way to mash up the VLOOKUP and MATCH formulas to get what we want. We can write `=VLOOKUP(valSalesPerson,$B$5:$N$17,MATCH(valMonth,lstMonth,0)+1,FALSE)` to fetch the value for a corresponding month.

This works by first finding the column number for the month we want with `MATCH(valMonth, lstMonth, 0)` portion. Then we just get value from corresponding column using VLOOKUP.

Why are we adding +1 to MATCH value?

By now you should be able to figure this out. So I am not explaining this part.

So which 2way lookup technique is best?

All 4 techniques have their merits. But my favorite is INDEX+MATCH technique. It is very simple to use and works very fast.

I also find the idea of using space operator to get the value at intersection almost magical. Every time I use it I feel like a rock star.

2nd time lucky? No... how about third time?

How to find 2nd or 3rd occurrence using lookup formulas

Looking up 2nd Occurrence in a List

(and 3rd, 4th etc...,)

Xxx	Yyy	Zzz	Abcd	Efgh
Xxx	Xxx	Xxx	Xxx	Xxx
Xxx	Xxx	Xxx	Xxx	Xxx
Xxx	Xxx	Xxx	Xxx	Xxx
Xxx	Xxx	Xxx	Xxx	Xxx
Xxx	Xxx	Xxx	Xxx	Xxx
Xxx	Xxx	Xxx	Xxx	Xxx
Xxx	Xxx	Xxx	Xxx	Xxx
Xxx	Xxx	Xxx	Xxx	Xxx

Using VLOOKUP we can get first match.

But what about getting the 2nd match
and corresponding items?

We know that VLOOKUP formula is useful to fetch the first matching item from a list. So what would you do if you need 2nd (or 3rd etc.) matching item from a list?

For eg. If you have below data, and you want to find out how much sales John made 2nd time, then VLOOKUP formula becomes quite useless. Or is it!?

Sales Person	No. Custom	Net Sales	Profit / Loss
Joseph	8	1,592	563
John	8	1,088	397
Josh	8	1,680	753
Jamie	9	2,133	923
Joseph	10	1,610	579
John	10	1,540	570
Josh	7	1,316	428
Jamie	7	1,799	709
Joseph	8	1,624	621
John	6	726	236
Josh	9	2,277	966
Jamie	6	714	221
John	9	2,682	1,023

Solution (with helper column)

A simple solution to this problem would be sorting our data on sales person's name. That way all Johns would line up one beneath another. And we just have to find the first John's position and add 1 to it to get to 2nd occurrence. Like this =MATCH("John", C5:C17, 0) + 1

But sorting is not an option all the time. ***So there should be a better way to do this?***

Well, there is. We just add a helper column before the sales person name and fill it with sales-person's name & occurrence. (see the below data table).

Helper Column	Actual Data			
SalesPerson #	Sales Person	No. Custom	Net Sales	Profit / Loss
Joseph1	Joseph	8	1,592	563
John1	John	8	1,088	397
Josh1	Josh	8	1,680	753
Jamie1	Jamie	9	2,133	923
Joseph2	Joseph	10	1,610	579
John2	John	10	1,540	570
Josh2	Josh	7	1,316	428
Jamie2	Jamie	7	1,799	709
Joseph3	Joseph	8	1,624	621
John3	John	6	726	236
Josh3	Josh	9	2,277	966
Jamie3	Jamie	6	714	221
John4	John	9	2,682	1,023

How to make the helper column?

For this we can use COUNTIF() Formula, like this: =C5&COUNTIF(\$C\$5:C5,C5).

Notice the \$C\$5:C5?, well the mix of absolute & relative references does the trick here and gets John1, John2... etc.

Now, to lookup 2nd occurrence of John, all we do is, simply write =VLOOKUP("John2",...) and we are done.

Bonus scenario – Get the last occurrence of John

One very common scenario is to find the last occurrence of John and return corresponding values (like sales etc.).

To find 2nd John, we searched for "John2". To find the last occurrence of John, we need to have *Johnx* where x=the number of times John appeared in the list.

This can be done by using another COUNTIF.

So we can use =VLOOKUP("John" & COUNTIF(\$C\$5:\$C\$17, "John"),)

The portion COUNTIF(\$C\$5:\$C\$17, "John") finds how many times John has appeared in our list of sales persons. Then "John" & COUNTIF(\$C\$5:\$C\$17, "John") would give *Johnx* where x is the count of Johns. In our case, it would give **John4**.

Should we always make a helper column?

While the above approach is excellent, it also requires that you set up and use an extra column of data. This may not be feasible for every situation. So how to find the sales of John2 without helper columns?

One approach – Using array formulas

I will confess first. This approach looks very complicated and mean. But it works very well. If you have struggled with array formulas presented in earlier chapters, you are going to hate this.

My suggestion?

Simple, read the technique, try to digest it. If it sounds all too complicated, live with helper column. When you have a genuine need and cannot afford a helper column, you can always re-read this section and understand how it works.

First let's look at the formula

Assuming that we want to get Sales (column E) for 2nd occurrence of John,

=INDEX(\$E\$5:\$E\$17, SMALL(IF(\$C\$5:\$C\$17="John", ROW(\$C\$5:\$C\$17)-ROW(\$C\$4)), 2))

Note: This is an array formula. So you need to press CTRL+Shift+Enter after typing this formula to get the result.

How does it work?

Let's start from inside out.

IF(\$C\$5:\$C\$17="John", ROW(\$C\$5:\$C\$17)-ROW(\$C\$4)) part

This checks column C for John and returns corresponding row number or FALSE. So the output will be,

{FALSE;2;FALSE;FALSE;FALSE;6;FALSE;FALSE;FALSE;10;FALSE;FALSE;13}

SMALL(IF(...), 2) part

This SMALL formula is going to tell us what is the 2nd smallest value in the list provided by IF().

So this becomes,

`SMALL({FALSE;2;FALSE;FALSE;FALSE;6;FALSE;FALSE;FALSE;10;FALSE;FALSE;13}, 2 )`

Hence the result would be 6.

What does this mean?

All this calculation simply means, the John we wanted is the 6th person in the list of sales people.

INDEX(\$E\$5:\$E\$17, SMALL(...)) part

The final piece of our puzzle is INDEX. This will give us the 6th item from sales column, since 6 is the result of SMALL(...) formula. Hence the answer will sales of 2nd occurrence of John.

Generic use of this formula

Assuming the name you are looking for is in A1, and the occurrence is in A2, you can use this generic version of the formula.

`=IFERROR(INDEX($E$5:$E$17,SMALL(IF($C$5:$C$17=$A$1,ROW($C$5:$C$17)-ROW($C$4)), $A$2)), "")`

The IFERROR part will prevent any error messages when you look for a name & occurrence combination that does not exist.

Some homework for you

Now that you are familiar with various techniques to get 2nd, 3rd or final occurrence of a value, let's test your skills. Write formulas to answer below questions:

- What is the Net Sales amount for last occurrence of Jamie?
- Of all the sales John made, which one is highest?
- Of all the sales John made, on which occurrence did he make the highest sale?

Lookup tips & gotchas

So far you have learned how to use VLOOKUP & various other formulas. Now let's talk about how to use these formulas effectively.

Here is a list of 5 tips that work wonders with VLOOKUP writing.

Use named ranges

No one likes a VLOOKUP with lots of dollars and cell references. They are unreadable and difficult to debug. So, a better option is to use named ranges in lookup formulas. For eg:
=VLOOKUP(valSalesPerson,tblData,3,FALSE) is easier to read and understand than
=VLOOKUP(G5,\$B\$5:\$G\$17,3,FALSE)

Make table / list references Absolute

When you need to write lookup formulas in a range of cells, the usual practice is to write first formula and then drag-fill. If you followed above advice and used named ranges, you should be ok with this approach. But if you do use cell references, make sure the table references are absolute, like this: \$B\$5:\$G\$17 instead of B5:G17. Here is a [nice tutorial explaining the concept of cell references](#).

Use Tables or Lists

Using named ranges or absolute references is good technique. But they suffer from one nagging limitation. If your source data (where you are looking up) grows or shrinks, you need to adjust the references. A better option? Use tables.

[Refer to chapter 3](#) for information on how to use tables.

Check for Errors

VLOOKUP is a powerful formula, alright. But ask it to look up a value that is not the data and it acts up. So you need to handle this. The easiest method is to use IFERROR() formula. Like this:
=IFERROR(VLOOKUP(...),"Oops, nothing found!").

But, IFERROR is a new formula in Excel 2007, so if you are using 2003 or before versions you need to use ISERROR (), like this: =IF(ISERROR(VLOOKUP(...)),"Oops, nothing found!",VLOOKUP(...))

But, there is another problem. The ISERROR() is not only lengthy, it is also ugly and inefficient as it calculates VLOOKUP twice. So a better approach is to use COUNTIF(), like this: =IF(COUNTIF(column, value you want to lookup)>0,VLOOKUP(...),"Oops, nothing found!") [more: [how to handle formula errors](#)]

Use VLOOKUP only when you need it

Do you know that formulas like SUMIF() or SUMPRODUCT() can effectively replace VLOOKUP() formulas? For eg. the formula =SUMIF(lookup-range, lookup value, return column range) gives same value as =VLOOKUP(lookup value, total range, 2, false) ? [Assuming there is only one match, return column range has numbers]

Similarly, if you just want to find whether a value is in a list or not, use COUNTIF() formula.

That is right. Not only formulas like SUMIF are better, they require no separate error handling. If the value cannot be found, they just return 0.

How to learn more about Excel?

If you are done reading and asking for more, I have good news for you. Visit below links for learning about all sorts of awesome things that you can do in Excel.

Visit our website & learn for free

Our mission at Chandoo.org is simple. *We want to make you awesome in your work.* We do this by sharing tips, tricks, tutorials, downloads, videos, examples & case studies on awesome uses of Excel, every week. Started in 2008, to date we have published more than 500 such resources on our site. And the good news is, all of this is free for you to grab. Just visit <http://chandoo.org/wp/> and start digging in.

[Chandoo.org website](http://chandoo.org/wp/)

[Join our Free Excel Tips Newsletter](#)

Join our online classes

Since our mission is to make you awesome, we also offer online classes so that you can learn in an intense, step-by-step fashion and get there faster. To date we have trained more than 5,000 professionals all over the world thru our online classes. We offer 3 types of classes. Choose your pick and become awesome.

Excel School Online Training

This program teaches you from fundamentals to advanced Excel so that you can create analyze data like a pro, make amazing charts, prepare complex models and come up with insightful dashboards. The course features more than 32 hours of video content that can be consumed at your own pace. You can download workbooks, presentations and video materials for practice or future reference. You can ask questions, interact with fellow classmates and learn from our online classroom 24x7.

[**Click here to visit Excel School program page to know more and enroll.**](#)

VBA & Excel Macro Classes

This program teaches you how to automate your work thru Excel VBA. You will learn how to generate your analysis, charts and reports with a click of a button in this program. The course features more than 24 hours of video training material along with 50 workbooks and example macros (VBA Code). It is a self-paced program and you can join and go thru the course any time.

[**Click here to visit VBA Classes page to know more and enroll.**](#)

Power Pivot & Advanced Data Analysis Class

Nowadays analysts everywhere are drowning in lots of data. Big data, the newest trend is nothing but using all your business data to understand what is going on and make smarter decisions. That is what our newest course teaches. It shows you how to use the most powerful feature of Excel that you have never heard. It is called **Power Pivot**. Available since Excel 2010, this very powerful feature can help you connect, analyze & visualize massive amounts of data (more than a million rows, really!!!) using Excel. It is like a full-fledged business intelligence (BI) solution in your Excel window.

Our course has more than 35 hours of video training material that can help you master Power Pivot. You can watch the lessons, download example workbooks and practice anytime from our online classroom.

[Click here to visit Power Pivot page to know more and enroll.](#)

Appendix 1: INDEX formula explained

INDEX formula plays a very important role when you are writing a lot of lookup formulas. So it makes sense to have a good grip on how INDEX works.

[Read this to understand INDEX Formula better.](#)

Appendix 2: Excel 2013 data model

Starting Excel 2013, Microsoft introduced a powerful feature called Data Model. This will help you connect different tables using keys (just like databases) and eliminates the need for VLOOKUP or other lookup formulas in report creation.

[See this to understand how Data Model can help you](#)

Thank you for taking time to become awesome

Thank you so much for taking time to improve yourself and becoming awesome.

May I request you to take 2 more minutes and share your review of this book thru Amazon so that more people can become awesome like you.

[Click here to post your review.](#)

Or just visit - <http://goo.gl/7T7SwN>

(This is case sensitive)

If you are amazon shy, you can always email me your review. Simply write to

chandoo.d@gmail.com

Thanks again.

Stay awesome,
Chandoo

"This is a book for everyone who uses Vlookup" - Dr. Nitin

the vlookup book

*Learn how to answer any question about
your data using Microsoft Excel lookup functions*



CHANDOO